

Numerical Calculus with Excel[®]

Habibollah Y. Far
Professor of Mathematics
Montgomery College
3200 College Park Dr. Conroe, Texas 77384
habib.y.far@nhmccd.edu

Abstract: Numerical calculus requires either graphing calculators or software. Calculators can carry these methods recursively or by using programs. Mathematical software also require programming knowledge or pre-written programs. Neither will help students to understand the mechanic of these methods. Widely available Microsoft Excel can carry these methods by writing small formulae and click and drag.

In this paper we take advantage of the intrinsic functions and cell reference capabilities of Microsoft Excel to demonstrate numerical calculus. Topics include fitting a linear model to a data, finding limits numerically, Riemann sums, Numerical integration, Newton's Method, Euler's Method, and Monte Carlo Integration.

Introduction: Numerical calculus requires either a graphing calculator or mathematical software. Most graphing calculators are not capable of doing what is needed on numerical methods unless they are done recursively or by using programs. Many mathematical software (Computer Algebra Systems, CAS) also require the knowledge of programming specifically for the software. Students can also use pre-written programs, but this doesn't help them to understand the mechanics of these methods. Widely available Microsoft Excel can be used for numerical calculus by writing a few small formulae and clicking and dragging. In this workshop we take advantage of intrinsic functions and cell reference capabilities of MS Excel to demonstrate instruction, computation, and visualization in numerical calculus.

Evaluation and Graphing of Functions: Any graphing calculator or CAS will graph functions by plotting points over an interval by dividing that interval into some subintervals, evaluating the function at some point over those subintervals, and plotting them. The plots are then connected to create the graph. Excel plots the graphs using the same technique.

Suppose we need to graph $f(x) = xe^{-x^2}$ over $[-3, 3]$. Divide $[-3, 3]$ into 12 subintervals of equal width. That is, let $\Delta x = (b - a) / n = 6 / 12 = 0.5$. Then, starting with $x = -3$, calculate $f(-3 + 0.5 * i)$ for $i = 0..12$. Finally: using Excel's Chart, plot and graph f .


- Enter i , x , and $f(x)$ in cells A1, B1, and C1 for column headings.
- Enter 0 and 1 in cell A2 and A3 respectively.
- Fill cells B2 and C2 with given formulae as in the following table.
- Select cells A2:A3.



- Click on the cell A3 box holder, hold and drag to cell A14.
- Cells A2:A14 will be filled with numbers 0 through 12.
- Select cells B2:C2.
- Click on the cell C2 box holder, hold and drag to cell C14. C2:C14 will be filled with the values of the function $f(x) = xe^{-x^2}$ for values in B2:B14. The second table is the result.


	A	B	C
1	<i>i</i>	<i>x</i>	<i>f(x)</i>
2	0	=-3+0.5*A2	=B2*EXP(-(B2^2))
3	1	=-3+0.5*A3	=B3*EXP(-(B3^2))
4	2	=-3+0.5*A4	=B4*EXP(-(B4^2))
5	3	=-3+0.5*A5	=B5*EXP(-(B5^2))
6	4	=-3+0.5*A6	=B6*EXP(-(B6^2))
7	5	=-3+0.5*A7	=B7*EXP(-(B7^2))
8	6	=-3+0.5*A8	=B8*EXP(-(B8^2))
9	7	=-3+0.5*A9	=B9*EXP(-(B9^2))
10	8	=-3+0.5*A10	=B10*EXP(-(B10^2))
11	9	=-3+0.5*A11	=B11*EXP(-(B11^2))
12	10	=-3+0.5*A12	=B12*EXP(-(B12^2))
13	11	=-3+0.5*A13	=B13*EXP(-(B13^2))
14	12	=-3+0.5*A14	=B14*EXP(-(B14^2))

	A	B	C
1	<i>i</i>	<i>x</i>	<i>f(x)</i>
2	0	-3	-0.00037023
3	1	-2.5	-0.00482614
4	2	-2	-0.03663128
5	3	-1.5	-0.15809884
6	4	-1	-0.36787944
7	5	-0.5	-0.38940039
8	6	0	0
9	7	0.5	0.38940039
10	8	1	0.36787944
11	9	1.5	0.15809884
12	10	2	0.03663128
13	11	2.5	0.00482614
14	12	3	0.00037023


To graph $f(x)$,

- select cells B1:C14.
- Click on the “Chart Wizard” icon .
- Select XY (Scatter), select continuous graph, and click next.
- Customize the graph and then click finish.





The result will be the following chart (graph).




To enhance the understanding of graphing concepts and to demonstrate graphing techniques such as stretching, shrinking, reflecting or translation, we can use the “Scrollbar”, or “slider” feature of the “Control Toolbox” under the “Toolbars” in Excel. This feature allows us to have a dynamic graph rather a static graph.

Suppose we need to show the effect of changes in a , b , c , and d in graph of $f(x) = a \sin(bx + c) + d$. Furthermore, suppose we need to graph $f(x) = a \sin x$ for $a = -2, -1.5, -1, -0.5, 0, 0.5, 1, 1.5$, and 2 over $[-2\pi, 2\pi]$ starting with $a = -2$, that is, graphing $f(x) = -2 \sin x$. For doing so, divide $[-2\pi, 2\pi]$ into 32 sub intervals with $\Delta x = 4\pi/32$. Since “Scrollbar” can take only nonnegative integers, enter 0 in C1 and link that cell to “slider” with a maximum value of 8. The value of a in cell B1 is $-2 + 0.5C1$, hence it satisfies our increments. As we move the “slider”, the values of f are recalculated and plotted. The first few lines of Excel output and its graph have been shown in next charts.

	A	B	C
1	a	=-2+C1*0.5	0
2	i	x	f(x)
3	0	=-2*PI()+A3*(4*PI()/32)	=\$B\$1*SIN(B3)
4	1	=-2*PI()+A4*(4*PI()/32)	=\$B\$1*SIN(B4)

	A	B	C
1	a	-2	0
2	i	x	f(x)
3	0	-6.28318531	-5E-16
4	1	-5.89048623	-0.7654



In order to insert a “Scrollbar”, after chart is complete,

- Right click somewhere on the Excel’s “Toolbar” and select “Control ToolBox”, or click on “View” and from the “Toolbars” menu select “Control ToolBox”.
- Click on “Design mode” icon in the “Control ToolBox” toolbar and select the scrollbar.
- Move the mouse onto the worksheet area, left click and drag to select a small rectangle.
- When the mouse button is released, the “slider” will appear.
- Right click on the slider and choose properties.
- Enter C1 in the “LinkedCell” of the properties dialog box.
- Set “Max” equal 8 and “Min” equal zero.
- Close the properties dialog box and click on the “Design Mode” icon again to exit.
- Moving the “slider” should change the value of C1 and consequently the graph.

To prevent the chart from resizing,

- Right click on one of the axis on the chart and choose “Format axis”.
- Select “Scale” tab.
- Uncheck the first four boxes under “Auto”.

Graphing $a \sin(x)$ is an example of change in amplitude or stretching, shrinking, or reflection. The Same procedure can be used to graph $\sin(bx)$, change in frequency or stretching/shrinking, $\sin(x+c)$, phase shift or horizontal translations, and $\sin(x)+d$, vertical translations.

Regression: In modeling, sometimes we need to find the equation of a line that best fits a set of data. In Excel we have a few options for achieving our goal. First, we can use “Regression”

command in the “Data Analysis” menu (package) of “Tools”. We can also use “Slope” and “Intercept” functions to find the slope and intercept of the line of best fit. Third, the slope and intercept can directly be calculated from the data using the regression equation $\hat{y} = b_0 + b_1x$ where $b_1 = \frac{n\sum xy - (\sum x)(\sum y)}{n\sum x^2 - (\sum x)^2}$ and $b_0 = \bar{y} - b_1\bar{x}$. The last method that we strongly suggest is trial-and-error. That is, let students guess the equation of the line of best fit. Then ask them to calculate the sum of squared error. The line with smallest sum of squared error is the best among all the guesses.

Suppose we need to find a line to fit the following data (ordered pair): $\{(1, 3.5), (2, 4.2), (3, 4.8), (4, 7.5), (5, 6.8), (6, 7.5)\}$.


- Enter the data in Excel.
- Plot the data as scattered plot using “Chart” and adjust the appearance of the chart as your liking. Guess the equation of the line and graph it on the same chart.
- Right click on the chart and select “Chart Options”. Click on the “Gridline” tabs and add the horizontal and vertical gridlines.

In order to add the second line to the graph,

- Right click on the chart and select “Source Data”.
- Choose the “Series” tab and add the second series.
- In the second series box enter $\{3, 8\}$.

	A	B	C	D
1			Guess an Equation	$y = ax+b$
2			a	b
3			1	2
4	Actual			
5	x	y	Y	(Err) ²
6	1	3.5	$=\$C\$3*A6+\$D\3	$=(B6-C6)^2$
7	2	4.2	$=\$C\$3*A7+\$D\3	$=(B7-C7)^2$
8	3	4.8	$=\$C\$3*A8+\$D\3	$=(B8-C8)^2$
9	4	7.5	$=\$C\$3*A9+\$D\3	$=(B9-C9)^2$
10	5	6.8	$=\$C\$3*A10+\$D\3	$=(B10-C10)^2$
11	6	7.5	$=\$C\$3*A11+\$D\3	$=(B11-C11)^2$
12				$=SUM(D6:D11)$

	A	B	C	D
1			Guess an Equation	$y = ax+b$
2			a	b
3			1	2
4	Actual Data			
5	x	y	Y	(Err) ²
6	1	3.5	3	0.25
7	2	4.2	4	0.04
8	3	4.8	5	0.04
9	4	7.5	6	2.25
10	5	6.8	7	0.04
11	6	7.5	8	0.25
12				2.87



After guessing an equation and plotting it, the predicted y values can be calculated. We can also calculate the sum of the squared errors for our guess and compare it with sum of the squared errors for other guesses in class. In our case, we have predicted $\hat{y} = 2 + x$. Sum of squared errors can be calculated using excel. Using the regression equation, as mentioned above, and Excel, we can calculate the slope and the intercept of the equation of the line of best fit.

	A	B	C	D	E
1	$b_0 =$	2	$b_1 =$	1	
2	x	y	y^{\wedge}	Error ²	Best fit err ²
3	1	3.5	=\$B\$1+\$D\$1*A3	=(B3-C3) ²	=((8/3)+0.8714*A3-B3) ²
4	2	4.2	=\$B\$1+\$D\$1*A4	=(B4-C4) ²	=((8/3)+0.8714*A4-B4) ²
5	3	4.8	=\$B\$1+\$D\$1*A5	=(B5-C5) ²	=((8/3)+0.8714*A5-B5) ²
6	4	7.5	=\$B\$1+\$D\$1*A6	=(B6-C6) ²	=((8/3)+0.8714*A6-B6) ²
7	5	6.8	=\$B\$1+\$D\$1*A7	=(B7-C7) ²	=((8/3)+0.8714*A7-B7) ²
8	6	7.5	=\$B\$1+\$D\$1*A8	=(B8-C8) ²	=((8/3)+0.8714*A8-B8) ²
9				=SUM(D3:D8)	=SUM(E3:E8)

	A	B	C	D	E
1	$b_0 =$	2	$b_1 =$	1	
2	x	y	y^{\wedge}	Error ²	Best fit err ²
3	1	3.5	3	0.25	0.001449071
4	2	4.2	4	0.04	0.043876284
5	3	4.8	5	0.04	0.231232751
6	4	7.5	6	2.25	1.816385138
7	5	6.8	7	0.04	0.050026778
8	6	7.5	8	0.25	0.156077671
9				2.87	2.299047693

	A	B	C	D
1	x	y	x ²	xy
2	1	3.5	1	3.5
3	2	4.2	4	8.4
4	3	4.8	9	14.4
5	4	7.5	16	30
6	5	6.8	25	34
7	6	7.5	36	45
8	21	34.3	91	135.3
9				
10	$\bar{x} =$	3.5	$\bar{y} =$	5.7167
11	$b_1 =$	0.8714	$b_0 =$	2.6667

	A	B	C	D
1	x	y	x ²	xy
2	1	3.5	=A2^2	=A2*B2
3	2	4.2	=A3^2	=A3*B3
4	3	4.8	=A4^2	=A4*B4
5	4	7.5	=A5^2	=A5*B5
6	5	6.8	=A6^2	=A6*B6
7	6	7.5	=A7^2	=A7*B7
8	=SUM(A2:A7)	=SUM(B2:B7)	=SUM(C2:C7)	=SUM(D2:D7)
9				
10	$\bar{x} =$	=A8/6	$\bar{y} =$	=B8/6
11	$b_1 =$	=(6*D8-A8*B8)/(6*C8-A8^2)	$b_0 =$	=D10-B11*B10

Limits: Suppose we need to find $\lim_{x \rightarrow 0} \frac{\sin x}{x}$ numerically by studying the behavior of its graph close to zero. We use Excel as follows to create a table and calculate the values of $\sin x/x$ close to zero, approaching from both sides. Excel will calculate $\sin x$ for x in radians. If x is in degrees, it has to be converted to radians. Also, as we realize Excel gives “#DIV/0” for $\sin 0/0$.

	A	B
1	x	f(x)
2	-1	=SIN(A2)/A2
3	-0.1	=SIN(A3)/A3
4	-0.01	=SIN(A4)/A4
5	-0.001	=SIN(A5)/A5
6	-0.00001	=SIN(A6)/A6
7	0	=SIN(A7)/A7
8	0.00001	=SIN(A8)/A8
9	0.0001	=SIN(A9)/A9
10	0.001	=SIN(A10)/A10
11	0.01	=SIN(A11)/A11
12	0.1	=SIN(A12)/A12
13	1	=SIN(A13)/A13

	A	B
1	x	f(x)
2	-1	0.84147098
3	-0.1	0.99833417
4	-0.01	0.99998333
5	-0.001	0.99999983
6	-0.00001	1
7	0	#DIV/0!
8	0.00001	1
9	0.0001	1
10	0.001	0.99999983
11	0.01	0.99998333
12	0.1	0.99833417
13	1	0.84147098

Numerical differentiation: Derivative of a function $f(x)$ at a point $(a, f(a))$, $f'(a)$, (slope of the tangent line to the graph of f at $(a, f(a))$) is given by $\lim_{h \rightarrow 0} ((f(a+h) - f(a))/h)$. Suppose we need to find the slope of the tangent line to the graph of $f(x) = \sin x/x$ at $x = \pi/4$. We can use the limit process to find this value.

	A	B	C	D
1		h	$a+h$	$(f(a+h)-f(a))/h$
2	0	=10^A2	=PI()/4+B2	=(SIN(C2)/C2-SIN(PI()/4)/(PI()/4))/B2
3	1	=10^A3	=PI()/4+B3	=(SIN(C3)/C3-SIN(PI()/4)/(PI()/4))/B3
4	2	=10^A4	=PI()/4+B4	=(SIN(C4)/C4-SIN(PI()/4)/(PI()/4))/B4
5	3	=10^A5	=PI()/4+B5	=(SIN(C5)/C5-SIN(PI()/4)/(PI()/4))/B5
6	4	=10^A6	=PI()/4+B6	=(SIN(C6)/C6-SIN(PI()/4)/(PI()/4))/B6
7	5	=10^A7	=PI()/4+B7	=(SIN(C7)/C7-SIN(PI()/4)/(PI()/4))/B7

	A	B	C	D
1		h	$a+h$	$(f(a+h)-f(a))/h$
2	0	1	1.79	-0.353065129
3	1	0.1	0.89	-0.259446374
4	2	0.01	0.8	-0.24736897
5	3	0.001	0.79	-0.246138935
6	4	0.0001	0.79	-0.246015714
7	5	1E-05	0.79	-0.24600339

Newton's Method: When finding exact real solution(s) to an equation, if they exist, is not possible, Newton's Method, which takes advantage of derivative and linearization, can be used to approximate these solutions. Suppose we need to find a zero of $f(x)$ over an interval. The process starts with an initial guess x_0 over that interval. If $f(x_0)$ is not within the tolerance level, then the next solution can be approximated by finding the intersection of the tangent line to the graph of f at $(x_0, f(x_0))$ and the x -axis. We can continue the process, if all requirements are met, recursively until $f(x)$ is close enough to x -axis for it to be called a zero. The equation of the tangent line to the graph of f at $(x_0, f(x_0))$ is given by $y = f'(x_0)(x - x_0) + f(x_0)$. Solving $f'(x_0)(x - x_0) + f(x_0) = 0$ (its intersection with x -axis) will results in $x = x_0 - f(x_0)/f'(x_0)$ or,

$$x_{n+1} = x_n - f(x_n)/f'(x_n).$$

Let $f(x) = x^5 + x - 1$ then $f'(x) = 5x^4 + 1$. Starting with initial guess $x_0 = 0$ we create the following Excel table to find zero of f . As it can be seen, we have our approximation in 4 or 5 steps depending upon the tolerance level.

	A	B	C	D
1	Step	x	f(x)	f(x)
2	0	0	=B2^5+B2-1	=5*B2^4+1
3	1	=B2-C2/D2	=B3^5+B3-1	=5*B3^4+1
4	2	=B3-C3/D3	=B4^5+B4-1	=5*B4^4+1
5	3	=B4-C4/D4	=B5^5+B5-1	=5*B5^4+1
6	4	=B5-C5/D5	=B6^5+B6-1	=5*B6^4+1
7	5	=B6-C6/D6	=B7^5+B7-1	=5*B7^4+1
8	6	=B7-C7/D7	=B8^5+B8-1	=5*B8^4+1
9	7	=B8-C8/D8	=B9^5+B9-1	=5*B9^4+1

	A	B	C	D
1	Step	x	f(x)	f(x)
2	0	0	-1	1
3	1	1	1	6
4	2	0.8333333	0.235211	3.411265
5	3	0.764382	0.025329	2.706916
6	4	0.755025	0.000386	2.624857
7	5	0.754878	9.32E-08	2.62359
8	6	0.754878	5.55E-15	2.62359
9	7	0.754878	0	2.62359

Newton's Method might fail for some functions: when the initial guess could cause the process to diverge, derivative at some point is zero, or the initial guess will give a zero which was not intended.

Newton's method can also be used to approximate irrational numbers such as π and $\sqrt{2}$.

	A	B	C	D
1	Step	x	f(x)	f(x)
2	0	3	=TAN(B2)	=(1/(COS(B2)))^2
3	1	=B2-C2/D2	=TAN(B3)	=(1/(COS(B3)))^2
4	2	=B3-C3/D3	=TAN(B4)	=(1/(COS(B4)))^2
5	3	=B4-C4/D4	=TAN(B5)	=(1/(COS(B5)))^2

	A	B	C	D
1	Step	x	f(x)	f(x)
2	0	3	-0.142546543	1.020319517
3	1	3.1397077491	0.001884907	1.000003553
4	2	3.1415926491	-4.46454E-09	1
5	3	3.1415926536	-1.22515E-16	1

	A	B	C	D
1	Step	x	f(x)	f(x)
2	0	1	=B2^2-2	=2*B2
3	1	=B2-C2/D2	=B3^2-2	=2*B3
4	2	=B3-C3/D3	=B4^2-2	=2*B4
5	3	=B4-C4/D4	=B5^2-2	=2*B5
6	4	=B5-C5/D5	=B6^2-2	=2*B6
7	5	=B6-C6/D6	=B7^2-2	=2*B7

	A	B	C	D
1	Step	x	f(x)	f(x)
2	0	1	-1	2
3	1	1.5	0.25	3
4	2	1.4166667	0.0069444	2.8333333
5	3	1.4142157	6.007E-06	2.8284314
6	4	1.4142136	4.511E-12	2.8284271
7	5	1.4142136	0	2.8284271

Riemann Sum: The area of a planar region lying between the graph of $f(x)$ and the x -axis can be approximated by calculating the sum of rectangular areas under the curve and x -axis. Suppose we need to approximate the area between the graph of $f(x)$, x -axis, $x = a$, and $x = b$, assuming $f(x) \geq 0$ for all x in $[a, b]$. The process starts by dividing the interval $[a, b]$ into n subintervals, that is, the width of each subinterval is given by $\Delta x = (b - a) / n$. Then we evaluate $f(a + i\Delta x)$, for $i = 0, 1, \dots, n - 1$ (if we start with the left-end point of each subinterval, otherwise, $i = 1, 2, \dots, n$). The approximation to the area then is obtained by adding the area of all rectangles width Δx and height $f(a + i\Delta x)$. That is, the approximate area is $\sum_{i=0}^{n-1} f(a + i\Delta x)\Delta x$.

Suppose we need to approximate the area between the graph of $f(x) = x^2 + 1$ and x -axis over the interval $[0, 2]$. Let $n = 8$. Then $\Delta x = (2 - 0) / 8 = 1/4$. The left-end point, the right-end point, and the midpoint approximation is given in the following Excel table.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1			a=	0		b=	2	n=	8		dx=	0.25		
2														
3	Left-end point approximation				Right-end point approximation				Mid point approximation					
4	n	x_i	$f(x_i)$	$f(x_i)*dx$	n	x_i	$f(x_i)$	$f(x_i)*dx$	n	x_i	$f(x_i)$	$f(x_i)*dx$		
5	0	0	1	0.25	1	0.25	1.0625	0.265625	0	0.125	1.015625	0.2539063		
6	1	0.25	1.0625	0.265625	2	0.5	1.25	0.3125	1	0.375	1.140625	0.2851563		
7	2	0.5	1.25	0.3125	3	0.75	1.5625	0.390625	2	0.625	1.390625	0.3476563		
8	3	0.75	1.5625	0.390625	4	1	2	0.5	3	0.875	1.765625	0.4414063		
9	4	1	2	0.5	5	1.25	2.5625	0.640625	4	1.125	2.265625	0.5664063		
10	5	1.25	2.5625	0.640625	6	1.5	3.25	0.8125	5	1.375	2.890625	0.7226563		
11	6	1.5	3.25	0.8125	7	1.75	4.0625	1.015625	6	1.625	3.640625	0.9101563		
12	7	1.75	4.0625	1.015625	8	2	5	1.25	7	1.875	4.515625	1.1289063		
13				4.1875				5.1875						4.65625
14														
15						Actual area=14/3=4.6667								

	A	B	C	D	E	F	G	H	I
1	a=	0	b=	2		n=	8	dx=	=(D1-B1)/G1
2									
3	n	x_i	$f(x_i)$	$f(x_i)*dx$		n	x_i	$f(x_i)$	$f(x_i)*dx$
4	0	=0+A4*\$I\$1	=B4^2+1	=C4*\$I\$1		1	=0+F4*\$I\$1	=G4^2+1	=H4*\$I\$1
5	1	=0+A5*\$I\$1	=B5^2+1	=C5*\$I\$1		2	=0+F5*\$I\$1	=G5^2+1	=H5*\$I\$1
6	2	=0+A6*\$I\$1	=B6^2+1	=C6*\$I\$1		3	=0+F6*\$I\$1	=G6^2+1	=H6*\$I\$1
7	3	=0+A7*\$I\$1	=B7^2+1	=C7*\$I\$1		4	=0+F7*\$I\$1	=G7^2+1	=H7*\$I\$1
8	4	=0+A8*\$I\$1	=B8^2+1	=C8*\$I\$1		5	=0+F8*\$I\$1	=G8^2+1	=H8*\$I\$1
9	5	=0+A9*\$I\$1	=B9^2+1	=C9*\$I\$1		6	=0+F9*\$I\$1	=G9^2+1	=H9*\$I\$1
10	6	=0+A10*\$I\$1	=B10^2+1	=C10*\$I\$1		7	=0+F10*\$I\$1	=G10^2+1	=H10*\$I\$1
11	7	=0+A11*\$I\$1	=B11^2+1	=C11*\$I\$1		8	=0+F11*\$I\$1	=G11^2+1	=H11*\$I\$1
12				=SUM(D4:D11)					=SUM(I4:I11)
13									
14							Actual area=	14/3=4.6667	

	K	L	M	N
2		Mid point	Approximation	
3	n	x_i	$f(x_i)$	$f(x_i)*dx$
4	0	=(1/8)+K4*\$I\$1	=L4^2+1	=M4*\$I\$1
5	1	=(1/8)+K5*\$I\$1	=L5^2+1	=M5*\$I\$1
6	2	=(1/8)+K6*\$I\$1	=L6^2+1	=M6*\$I\$1
7	3	=(1/8)+K7*\$I\$1	=L7^2+1	=M7*\$I\$1
8	4	=(1/8)+K8*\$I\$1	=L8^2+1	=M8*\$I\$1
9	5	=(1/8)+K9*\$I\$1	=L9^2+1	=M9*\$I\$1
10	6	=(1/8)+K10*\$I\$1	=L10^2+1	=M10*\$I\$1
11	7	=(1/8)+K11*\$I\$1	=L11^2+1	=M11*\$I\$1
12				=SUM(N4:N11)

Increasing the number of subintervals to 100 will refine this approximation to 4.628.

The actual area can be obtained analytically by letting the number of subintervals approach infinity. That is $\text{area} = \lim_{n \rightarrow \infty} \sum_{i=1}^n f(c_i)\Delta x$, where $x_{i-1} \leq c_i \leq x_i$. In addition, c_i 's can be anywhere on the interval $[x_{i-1}, x_i]$. We can take advantage of the Rand() function in Excel to show this fact.

	A	B	C	D
1	a=	=0	b=	=2
2	n=	8	dx=	=(D1-B1)/B2
3				
4	n	x_i	$f(x_i)$	$x_i*f(x_i)$
5	0	=A5*\$D\$2+RAND()/4	=B5^2+1	=\$D\$2*C5
6	1	=A6*\$D\$2+RAND()/4	=B6^2+1	=\$D\$2*C6
7	2	=A7*\$D\$2+RAND()/4	=B7^2+1	=\$D\$2*C7
8	3	=A8*\$D\$2+RAND()/4	=B8^2+1	=\$D\$2*C8
9	4	=A9*\$D\$2+RAND()/4	=B9^2+1	=\$D\$2*C9
10	5	=A10*\$D\$2+RAND()/4	=B10^2+1	=\$D\$2*C10
11	6	=A11*\$D\$2+RAND()/4	=B11^2+1	=\$D\$2*C11
12	7	=A12*\$D\$2+RAND()/4	=B12^2+1	=\$D\$2*C12
13				=SUM(D5:D12)
14				

	A	B	C	D
1	a=	0	b=	2
2	n=	8	dx=	0.25
3				
4	n	x_i	$f(x_i)$	$f(x_i)*dx$
5	0	0.173506	1.030105	0.25752613
6	1	0.310124	1.096177	0.2740442
7	2	0.529171	1.280022	0.32000556
8	3	0.793943	1.630346	0.40758644
9	4	1.118665	2.251411	0.56285283
10	5	1.461582	3.136222	0.78405546
11	6	1.749307	4.060077	1.01501913
12	7	1.751224	4.066785	1.01669613
13				4.63778588
14	Actual area=14/3=4.667			

The Riemann sum relaxes the requirement for the width of subintervals to be the same. The area can be found by $\sum_{i=1}^n f(c_i)\Delta x_i$ where $x_{i-1} \leq c_i \leq x_i$ and Δx_i is the width of i th subinterval.


	A	B	C	D
1	n	x_i	$f(x_i)$	$f(x_i)*dx$
2	1	=RAND()/4	=B2^2+1	=B2*C2
3	2	=B2+RAND()/4	=B3^2+1	=(B3-B2)*C3
4	3	=B3+RAND()/4	=B4^2+1	=(B4-B3)*C4
5	4	=B4+RAND()/4	=B5^2+1	=(B5-B4)*C5
6	5	=B5+RAND()/4	=B6^2+1	=(B6-B5)*C6
7	6	=B6+RAND()/4	=B7^2+1	=(B7-B6)*C7
8	7	=B7+RAND()/4	=B8^2+1	=(B8-B7)*C8
9	8	=B8+RAND()/4	=B9^2+1	=(B9-B8)*C9
10	9	=B9+RAND()/4	=B10^2+1	=(B10-B9)*C10
11	10	=B10+RAND()/4	=B11^2+1	=(B11-B10)*C11
12	11	=B11+RAND()/4	=B12^2+1	=(B12-B11)*C12
13	12	=B12+RAND()/4	=B13^2+1	=(B13-B12)*C13
14	13	=B13+RAND()/4	=B14^2+1	=(B14-B13)*C14
15	14	=B14+RAND()/4	=B15^2+1	=(B15-B14)*C15
16	15	=B15+RAND()/4	=B16^2+1	=(B16-B15)*C16
17			area=	=SUM(D2:D16)

	A	B	C	D
1	n	x_i	$f(x_i)$	$f(x_i)*dx$
2	1	0.02212	1.00049	0.02213
3	2	0.08035	1.00646	0.05860
4	3	0.19629	1.03853	0.12041
5	4	0.43308	1.18756	0.28121
6	5	0.55625	1.30941	0.16127
7	6	0.71179	1.50665	0.23435
8	7	0.83809	1.70240	0.21501
9	8	1.03244	2.06593	0.40152
10	9	1.05055	2.10366	0.03810
11	10	1.20447	2.45074	0.37721
12	11	1.38268	2.91179	0.51891
13	12	1.61744	3.61611	0.84893
14	13	1.79121	4.20844	0.73131
15	14	1.96220	4.85022	0.82931
16	15	1.99407	4.97630	0.15859
17			area=	4.99686

In our calculation we reach the result with 15 subintervals and area of 4.99686.

Monte Carlo Method: The Monte Carlo method relies on the random numbers to approximate the area of a region between a function f and the x-axis (or another function).

Suppose we need to approximate the area bounded by the graph of $f(x) = e^{(1-x)^3}$, x -axis, $x = 0$, and $x = 2$, as shown in the opposing graph. That is, we need to evaluate $\int_0^2 e^{(1-x)^3} dx$. First, we select a random number, x , on $[0, 2]$ and evaluate $f(x)$. Next, we select another random number, X , between 0 and 3. If $f(x) \leq X$, then the selected point is above the curve, in the rectangular area of size 2×3 ;



otherwise, it is under (or on) the curve. The ratio of the number of points under the curve to the total number of points is the area under the curve. The approximation improves as the number of selected random points increases.

In Excel, the function Rand() selects a decimal number on $[0, 1]$ randomly. Since we are approximating the area over $[0, 2]$, then we multiply each random number by 2 for x -axis. Since, we are comparing the area under the curve with the area of the rectangular region of height 3 then the second selected random number will be multiplied by 3. Next, we use IF function to decide whether our random points are above or below the curve. Rand() recalculates every time a function is executed in Excel. After all calculations are done, we can use the Copy, Paste Special feature of Excel to change all formulae to numeric values (all formulae will be deleted.)

	A	B	C	D	E
1	No.	x	$f(x)$	X	Test
2	1	=2*RAND()	=EXP((1-B2)^3)	=3*RAND()	=IF(C2<=D2,0,1)
3	=A2+1	=2*RAND()	=EXP((1-B3)^3)	=3*RAND()	=IF(C3<=D3,0,1)
4	=A3+1	=2*RAND()	=EXP((1-B4)^3)	=3*RAND()	=IF(C4<=D4,0,1)
5	=A4+1	=2*RAND()	=EXP((1-B5)^3)	=3*RAND()	=IF(C5<=D5,0,1)
6	=A5+1	=2*RAND()	=EXP((1-B6)^3)	=3*RAND()	=IF(C6<=D6,0,1)

In our calculations, the number of points under the curve (or on the curve) were 36 which gives us the area of $6(36/100) = 2.16$. The Maple approximates the area as 2.1494.

The Monte Carlo Method can also be used for double or triple integrals.

	A	B	C	D	E
1	No.	x	$f(x)$	X	Test
2	1	1.242088	0.985912	1.448012	0
3	2	1.042912	0.999921	1.294691	0
4	3	1.193031	0.992833	1.763505	0
5	4	0.679147	1.033582	0.489309	1
6	5	0.091314	2.11766	1.822992	1
7	6	0.162079	1.80095	0.299494	1
8	7	1.31102	0.970362	2.930291	0
9
10
11
12	100	1.998253	0.369809	1.903093	0
13	Number of points under curve				36

Euler’s Method (Numerical Differential Equations): Euler’s Method, as Newton’s Method relies on linearization to approximate the solution y of a differential equation (initial value problem) $y' = f(x, y)$ with initial condition $y(x_0) = y_0$ at some x . The equation of the tangent line to y at (x_0, y_0) , using a very small increment dx , is given by $y = y_0 + f(x_0, y_0)dx$. Recursive calculations of (x_i, y_i) will approximate $y = y(x)$.


Suppose we need to approximate the solution of $y' = y - e^{2x}$, with $y(0) = 1$, $dx = 0.25$ at $x = 2$.

We use Excel to approximate y_i and compare it with exact solution $y = -e^{2x} + 2e^x$.

	A	B	C
1	x	y (Euler)	y (Excat)
2	0	1	$=-EXP(2*A2)+2*EXP(A2)$
3	0.25	$=B2+(B2-EXP(2*A2))*0.25$	$=-EXP(2*A3)+2*EXP(A3)$
4	0.5	$=B3+(B3-EXP(2*A3))*0.25$	$=-EXP(2*A4)+2*EXP(A4)$
5	0.75	$=B4+(B4-EXP(2*A4))*0.25$	$=-EXP(2*A5)+2*EXP(A5)$
6	1	$=B5+(B5-EXP(2*A5))*0.25$	$=-EXP(2*A6)+2*EXP(A6)$
7	1.25	$=B6+(B6-EXP(2*A6))*0.25$	$=-EXP(2*A7)+2*EXP(A7)$
8	1.5	$=B7+(B7-EXP(2*A7))*0.25$	$=-EXP(2*A8)+2*EXP(A8)$
9	1.75	$=B8+(B8-EXP(2*A8))*0.25$	$=-EXP(2*A9)+2*EXP(A9)$
10	2	$=B9+(B9-EXP(2*A9))*0.25$	$=-EXP(2*A10)+2*EXP(A10)$

	A	B	C
1	x	y (Euler)	y (Excat)
2	0	1	1
3	0.25	1	0.919329563
4	0.5	0.837819682	0.579160713
5	0.75	0.367704146	-0.24768904
6	1	-0.66079209	-1.95249244
7	1.25	-2.67325413	-5.20180805
8	1.5	-6.38719115	-11.1221588
9	1.75	-13.0053732	-21.6062466
10	2	-24.5355795	-39.8200378

The Euler's method results in -24.5358 , where $y(2) = -39.82$. We can also look at their graphs.



Conclusion: Excel is a great tool for calculation and presentation in teaching numerical calculus. In addition to its use in any of the methods mentioned above, it can be used for other numerical calculus techniques such as volume, arc length, line integral, etc. It can also be used in Statistics and Finite Math classes.

References:

Larson, Hostetler, & Edwards, *Calculus: Early Transcendental Functions*, 3rd Edition, Houghton Mifflin Company (2001).

Finny, Weir, & Giordano, *Calculus, Early Transcendentals*, 10th Edition, Addison-Wesley (2000).

Stewart, *Calculus, Early Transcendentals*, 5th Edition, Thomson, Brooks/Cole (2002)..